title: Import to Specific Excel Sheet from Access via VBA
body: <p>I am trying to figure out how to get data that I am importing into Excel from an Access table to import into a specific sheet (Either a sheet just called Sheet 2 or Access Data). I have the following code to get the data and to format it the way I want after import, but I can't get it to import into a specific sheet. Can I get assistance? Here's what I have:</p>
  
  <p>Update to the code with resolution: </p>
  
  <pre><code>Sub getAccessData()
  
  Dim DBFullName As String
  Dim Connect As String, Source As String
  Dim Connection As ADODB.Connection
  Dim Recordset As ADODB.Recordset
  Dim Col As Integer
  Dim lngLastColumn As Long
  Dim lngLastRow As Long
  Dim OXLSheet As Worksheet
  
  Set OXLSheet = Worksheets("WorksheetName")
  
  Worksheets("WorksheetName").Cells.Clear
  
  'Datebase path info
  DBFullName = "C:\Users\myname\Desktop\Database Backups\database.accdb"
  
  'Open the connection for the database
  Set Connection = New ADODB.Connection
  Connect = "Provider=Microsoft.ACE.OLEDB.12.0;"
  Connect = Connect &amp; "Data Source=" &amp; DBFullName &amp; ";"
  Connection.Open ConnectionString:=Connect
  
  
  'Create RecordSet
  Set Recordset = New ADODB.Recordset
  With Recordset
  
      'Data Filter
      Source = "SELECT * FROM tblRetirements WHERE [AllowEnteredInPayroll] Is Null AND ApplicationCancelled = 'No'"
      .Open Source:=Source, ActiveConnection:=Connection
  
  
      'Write field Names
      For Col = 0 To Recordset.Fields.Count - 1
          Worksheets("WorksheetName").Range("A5").Offset(0, Col).Value = Recordset.Fields(Col).Name
      Next
  
      'Write Recordset
      Worksheets("WorksheetName").Range("A5").Offset(1, 0).CopyFromRecordset Recordset
  End With
  ActiveSheet.Columns.AutoFit
  Set Recordset = Nothing
  Connection.Close
  Set Connection = Nothing
  
  
  
  With OXLSheet
      lngLastColumn = .Cells(5, .Columns.Count).End(xlToLeft).Column
      lngLastRow = .Cells(.Rows.Count, "A").End(xlUp).Row
      .ListObjects.Add(xlSrcRange, .Range(.Cells(5, 1), .Cells(lngLastRow, lngLastColumn)), , xlYes).Name = "Table1"
  
      ActiveSheet.ListObjects("Table1").TableStyle = "TableStyleMedium16"
  End With
  
  End Sub
  </code></pre>
  
  <p>Thank you.</p>